


PC-1500 袖珍计算机 BASIC 语言应用程序集

PC-1500 袖珍计算机 BASIC 语言

应用程序集



TP312
101

前 言

PC-1500袖珍计算机,是当前国内最为流行的袖珍机型。由于该机价格低,性能强,有自备电源不怕断电,又便于携带,因此,它拥有大量用户。为了提高广大基层工程技术人员计算机软件的素质,普及微型计算机的应用,作者从近年实际工作所编制和使用的大量软件中,筛选出近百个应用程序编入此书。书中每个程序的数学模型介绍,力求精炼;而程序的操作指南,力求清楚;并附有详细的程序清单,以便使读者,拿来即可应用。由于这些应用程序都是用规范的BASIC语句编写的,因此,稍加修改即可在APPLE II、IBM PC等微型机上运行。

由于PC-1500的微型打印机CE-150,实际是一部X-Y绘图仪,因此,我们从图形输出的角度,提供了用PC-1500打印汉字的软件,近一步开拓了它的功能。该汉字软件不占字库,只对需要打印的汉字进行规格编码,即可从CE-150的窄条上输出,因此,可以大大节约内存。

本书的读者与使用对象,是气象、农林、水利、工程设计等科技人员,也可作为普及PC-1500袖珍机应用提高班的教材。该书的全部应用程序,可录制在五盒C-60普通盒式磁带上,或通过CE-158接口与联机软件,连接IBM PC微型机,录在一张双面双密度5吋软盘上。这些,可以大大方便广大的计算机用户。

本书所提供的PC-1500袖珍机应用程序,全部是用通讯软件先从PC-1500袖珍机传入IBM PC/XT磁盘,再用M2024宽行打印机打印的。由于IBM PC/XT的BASIC语言中,数值有单精、双精度之分,因此程序清单中个别的数字后面带有“!”号,键入PC-1500时,请注意应将“!”删去。另外,本程序集所提供的各类打印样品与图形输出,均可绘为彩色(黑、兰、绿、红四色),限于印刷条件,本书均印成黑色。

参加本书编辑并提供程序文本的有王宝成、弭明详等同志,张敏同志为协助此书的出版作了不少工作。本书在编写过程中,还得到天津市气象局研究所及科教处的于恩洪、芦至同志的大力支持,在此一并致谢。

鉴于作者水平有限,成书仓促,本书的错误和不足之处一定不少,敬希读者不吝赐教。

宛公展 1986年12月16日
于天津市气象科学研究所

目 录

第一章 基本功能模块	(1)
§ 1 线性代数计算	(1)
行列式求值(1-1-01)	(1)
矩阵乘法(1-1-02)	(2)
矩阵求逆(1-1-03)	(4)
迭代法解线性方程组(1-1-04)	(6)
高斯-约当消去法解线性方程组(1-1-05)	(8)
求实对称矩阵的特征值与特征向量(1-1-06)	(10)
用直接法求特征值(1-1-07)	(13)
施多姆-牛顿法解高次代数方程(1-1-08)	(16)
二分法求解超越方程(1-1-09)	(19)
§ 2 数据处理	(21)
求序列的平均值、标准差、变异系数与距平值(1-2-01)	(21)
序列标准化处理(1-2-02)	(22)
求序列的极值、极值位置,并顺序排队(1-2-03)	(23)
序列给权平滑处理(1-2-04)	(24)
自相关系数计算(1-2-05)	(26)
简单相关系数计算(1-2-06)	(27)
相关矩阵计算(1-2-07)	(28)
偏相关系数计算(1-2-08)	(29)
N点一次平滑(1-2-09)	(32)
N点二次平滑(1-2-10)	(33)
五点三次平滑(1-2-11)	(35)
§ 3 绘图与打印表格技巧	(37)
一般函数绘图(1-3-01)	(37)
极坐标与参数方程绘图(1-3-02)	(39)
拟合数据绘图(1-3-03)	(44)
绘星座图(1-3-04)	(46)
绘圆形百分率图(1-3-05)	(48)
绘直方图(1-3-06)	(49)
三变量点聚图(1-3-07)	(51)
通用打印表格输出(1-3-08)	(53)
§ 4 概率统计检验计算	(55)
统计参数——各阶矩的计算(1-4-01)	(55)
正态分布检验(1-4-02)	(55)
t 检验计算(1-4-03)	(55)
χ^2 检验计算(1-4-04)	(55)
F 检验计算(1-4-05)	(55)

§ 5 数据资料记带	(66)
数组记入磁带(1-5-01)	(66)
数组浓缩记带(1-5-02)	(67)
§ 6 汉字打印软件	(68)
PC-1500汉字打印程序(1-6-01)	(68)
第二章 应用程序	(73)
§ 1 气象统计预报	(73)
一元曲线回归(2-1-01)	(73)
多元回归(2-1-02)	(76)
逐段回归(1)(2-1-03)	(79)
逐段回归(2)(2-1-04)	(83)
正交筛选回归(2-1-05)	(86)
REEP回归(2-1-06)	(89)
0,1回归(2-1-07)	(92)
逐步回归(2-1-08)	(95)
自回归(2-1-09)	(98)
用方差分析找周期(2-1-10)	(100)
谐波分析(2-1-11)	(103)
功率谱分析(2-1-12)	(106)
类平均聚类分析(2-1-13)	(108)
模糊聚类分析(2-1-14)	(112)
单站常用物理量计算(2-1-15)	(116)
自然正交展开(2-1-16)	(120)
切比雪夫正交多项式展开(2-1-17)	(125)
积分回归(2-1-18)	(127)
岑回归(2-1-19)	(132)
§ 2 气象台站测报	(137)
本站气压订正简表(2-2-01)	(137)
海平面气压高度差订正简表(2-2-02)	(139)
毛发湿度订正图(2-2-03)	(140)
逐日日照时数查算表(2-2-04)	(144)
土壤湿度计算(2-2-05)	(146)
农业气象观测发报程序(2-2-06)	(148)
逐日气象要素时间演变曲线(2-2-07)	(1)
制作农气表-1程序(2-2-08)	(157)
§ 3 气象资料服务	(162)
气候要素综合图(2-3-01)	(162)
风玫瑰图(2-3-02)	(169)
地面粗糙度计算(2-3-03)	(172)
近地面风速垂直廓线计算(2-3-04)	(173)
用皮尔逊III型曲线求水文气象要素的概率分布(2-3-05)	(176)
小球测风资料整理计算(2-3-06)	(178)

用龚贝尔分布求气候要素极值(2-3-07).....	(180)
气表-1 风速计算(2-3-08).....	(183)
太阳高度角计算(2-3-09).....	(186)
§ 4 农业产量气象预测预报.....	(188)
产量资料记带(2-4-01).....	(188)
产量资料订正(2-4-02).....	(190)
产量聚类分区(2-4-03).....	(197)
预报因子的膨化相关普查筛选(2-4-04).....	(201)
农业产量前景的线性型动态监测模式(2-4-05).....	(206)
农业产量前景的非线性型动态监测模式(2-4-06).....	(214)
§ 5 系统分析初步.....	(221)
线性规划求解(2-5-01).....	(221)
灰色系统-关联矩阵计算(2-5-02).....	(226)
§ 6 气候资源分析.....	(228)
气候资源保证率计算与绘图(2-6-01).....	(228)
热量资源整编(2-6-02).....	(231)
降水资源整编(2-6-03).....	(240)
太阳辐射资料计算(2-6-04).....	(246)
气候资源数据记带专用程序(2-6-05).....	(249)
§ 7 业务管理与其它.....	(250)
工资管理程序(2-7-01).....	(250)
汉字工资程序(2-7-02).....	(254)
十功能电子钟(2-7-03).....	(260)
万年日历(2-7-04).....	(262)
PC-1500袖珍机验机程序(2-7-05).....	(266)
天气预报质量评分(2-7-06).....	(268)
第三章 BASIC 语言编程技巧.....	(276)
§ 1 相关矩阵计算程序的优化设计.....	(276)
§ 2 对称型矩阵的拉长存放技巧.....	(278)
§ 3 微型技巧集锦.....	(280)
§ 4 正交多项式的计算、存放与调用.....	(285)
§ 5 程序的合并运行.....	(287)
§ 6 恢复已被清除(NEW)的程序.....	(289)
第四章 PC-1500与IBM PC/XT的联机通信技术.....	(291)
§ 1 联机通信概述.....	(291)
§ 2 联机的硬件构成.....	(292)
§ 3 联机通信的项目与软件支持.....	(293)
§ 4 用PC-1500袖珍机建IBM PC/XT微型机DBASE数据库的方法.....	(301)
§ 5 通信软盘的制作.....	(304)

第一章 基本功能模块

§1 线性代数计算

程序名称：行列式求值

文件名：PRO 1-1-01 占用字节：479

编制者：宛公展

一、功能与方法概述

本算法可求出任意 n 阶行列式之值。计算时，首先在行列式上三角区域按行检查主元，如果主元为0，则继续沿着行搜索各列，当找到一个非0元素，便进行行列交换，使主元素换成非0。然后，按下式将对应的列元素消成0。变换公式为

$$a_{ij} = a_{ij} - a_{ik} \cdot a_{kj} / a_{kk}$$

这里， k 为变换的步数。当整个变换结束后，行列式的下三角区域，可全消成0，行列式之值，即为对角线各元素之积。

二、操作指南与举例

例如，欲求四阶行列式

$$\begin{vmatrix} 0 & 1 & 1 & 1 \\ 1 & 0 & 1 & 1 \\ 1 & 1 & 0 & 1 \\ 1 & 1 & 1 & 0 \end{vmatrix}$$

之值。操作步骤如下：

(一) 先在程序的500句，置入行列式的阶数 N 。本例 $N=4$ 。

(二) 从510句开始，按行将行列式各元素置入，本例依次为0,1,1,1,……1,1,1,0。

(三) 用RUN指令开工运行，液晶显示出计算结果为：DET=-3。

三、程序清单

```
5 REM PRO 1-1-01
6 REM PC--1500" !DET!=? "
10 CLEAR
20 READ N
30 DIM A(N,N)
40 FOR I=1 TO N:FOR J=1 TO N
50 READ A(I,J)
60 NEXT J:NEXT I
65 D=1
70 FOR L=1 TO N:FOR I=L TO N
80 IF A(I,L)<>0 GOTO 110
90 NEXT I
100 BEEP 20:PRINT " DET= 0":END
110 IF I=L GOTO 160
120 FOR J=1 TO N
130 B=A(L,J):A(L,J)=A(I,J):A(I,J)=B
140 NEXT J:D=-D
```

```

160 FOR K=L TO N
170 IF K=LGOTO 220
180 U=A(K,L)/A(L,L)
190 FOR J=L TO N
200 A(K,J)=A(K,J)-U*A(L,J)
210 NEXT J
220 NEXT K:BEEP 1
230 NEXT L
240 C=1:FOR K=1 TO N
250 C=C*A(K,K)
260 NEXT K
270 PRINT " DET=":C*D
300 END
500 DATA 4
510 DATA 0,1,1,1,1,0,1,1,1,1,0,1,1,1,1,0

```

程序名称：矩阵乘法

文件 名：PRO 1-1-02 占用字节：894

编制者：宛公展

一、功能与方法概述

本算法可作 $C = A * B$ 的矩阵乘法运算，其中 A 为 m 行 p 列， B 为 p 行 n 列， C 为 m 行 n 列。运算结果 C 即可通过液晶显示，也可根据预选字型在窄条上横向打印输出。当一个指定页打不下时，计算机可根据字号大小及矩阵 C 的行数，判断出应打印的页数，并自动进行换页打印。

二、操作指南与举例

例如，欲求 $C = A * B$ ，其中

$$A = \begin{pmatrix} 3 & -2 \\ 1 & 0 \\ -4 & 3 \\ -2 & 1 \end{pmatrix}; \quad B = \begin{pmatrix} 1 & 2 & 4 & 0 \\ 0 & 3 & 1 & 5 \end{pmatrix}$$

操作步骤如下：

(一) 先在程序的500句，置入基本参数 M, N, P ；本例分别为4,4,2。

(二) 然后在510与520句，分别将 A 与 B 的元素按行置入。

(三) 用指令 `RUN` 开工运行。当计算机提问：`LPRINT OR PRINT(L/P)?` 时，请回答 `L`（表示计算结果打印）或 `P`（结果只显示不打印）。

(四) 如果回答 `P`，当再提问 `TIME?` 时，请回答显示的时间间隔，如键入2，可使 C 矩阵的每个元素，在液晶上以2秒钟的间隔依次显示，余可类推。

(五) 如果回答 `L`，当再提问 `CSIZE?` 时，请键入打印字号（1—3号）。

注意：本例回答 `L`，可从窄条上将 C 矩阵打印出， $C = \begin{pmatrix} 3 & 0 & 10 & -10 \\ 1 & 2 & 4 & 0 \\ -4 & 1 & -13 & 15 \\ 2 & -1 & -7 & 5 \end{pmatrix}$

详见打印样品，样品以1号及2号字打印，虚线为分页符号，行号“1:，2:，……”、列号“1,2,……”，可用不同颜色打印，以便与矩阵元素相区别。

打印样品：

1:	1	2	3	4	1:	1	2	3	4
2:	3	0	10	-10	2:	3	0	10	-10
3:	1	2	4	0	3:	1	2	4	0
4:	-4	1	13	15	4:	-4	1	13	15
	-2	-1	-7	5		-2	-1	-7	5

三、程序清单

```

5 REM PRO 1-1-02
6 REM PC-1500" C=A*B"
10 CLEAR
20 READ M,N,P
30 DIM A(M,P),B(P,N),C(M,N)
40 FOR I=1 TO M:FOR J=1 TO P
50 READ A(I,J)
60 NEXT J:NEXT I
70 FOR I=1 TO P:FOR J=1 TO N
80 READ B(I,J)
90 NEXT J:NEXT I
95 INPUT "LPRINT OR PRINT (L/P)?":S$
100 IF S$="P"INPUT "TIME=?":T
110 WAIT 64*T
120 FOR I=1 TO M:FOR J=1 TO N
125 C=0
130 FOR K=1 TO P
135 C=C+A(I,K)*B(K,J)
140 NEXT K
150 C(I,J)=C
160 IF S$<>"P"GOTO 180
170 CLS:PRINT " C(" +STR$(I)+", "+STR$(J)+")= " :C
180 NEXT J:NEXT I
190 IF S$<>"L"THEN END
195 INPUT "CSIZE ?":CS
200 GRAPH:ROTATE 1:CSIZE CS
210 CN=24/CS
220 P=INT (M/CN):PL=M-P*CN
230 IF PL>0 LET P=P+1
240 FOR PA=1 TO P
245 Y=5*CS*7
250 LINE (0,-Y/2)-(210,-Y/2),8,3
255 I2=CN:IF PL>0 AND PA=PLET I2=PL
260 FOR J=1 TO N
270 FOR I=0 TO I2
275 X=(CN-I)/CN*200
280 IF J=1 AND I>0GOCURSOR (X,0):COLOR 2:LPRINT USIN
G "#####" :CN*(PA-1)+I:": "

```

```

284 IF I=0 LET E=J:COLOR 3:GOTO 295
285 E=C(CN*(PA-1)+I,J):COLOR 2
295 GLCURSOR (X,-Y):LPRINT USING "#####" E
300 NEXT I
305 IF J=NLET Y=Y+2
310 GLCURSOR (0,-Y):SORGN
320 NEXT J
330 NEXT PA
335 LINE (0,0)-(210,0),0,3
340 TEXT :LF 3
350 END
500 DATA 4,4,2
510 DATA 3,-2,1,0,-4,3,-2,1
520 DATA 1,2,4,0,0,3,1,5

```

程序名称：矩阵求逆

文件 名：PRO 1-1-03 占用字节：1312

编制者：宛公展

一、功能与方法概述

本程序采用列主元迭代消去法,可对 n 阶方阵 A 直接求出逆矩阵 A^{-1} 。其变换公式为:

$$\begin{cases} a_{kk} = 1/a_{kk} \\ a_{kj} = a_{kj} a_{kk} \\ a_{ij} = a_{ij} - a_{ik} a_{kj} \\ a_{ik} = -a_{ik} a_{kk} \end{cases}$$

上式中, k 为变换的步数,该变换的要点是,主元原位求倒数,主元所在行乘以主元,主元所在列乘主元并加负号,非主元行列的元素,实施正交变换。在消去过程中,若当步主元为 0,可沿着列搜索出一个非 0 元素,并进行行交换,使主元变为非 0,再用上式进行运算。迭代结束后,曾交换过的行再通过交换列使其还原。如果 A 阵为奇异矩阵,蜂鸣器自动报警,并显示无逆标志。

二、操作指南与举例

例如, $A = \begin{pmatrix} 0 & 1 & 2 \\ 3 & 2 & 1 \\ 4 & 3 & 5 \end{pmatrix}$, 求 $A^{-1} = ?$ 操作步骤如下:

(一) 在程序的 1000 句,置方阵 A 的阶数 N ,本例 $N = 3$ 。

(二) 从 1010 句开始,按行置入 A 的各个元素。

(三) 用 RUN 指令开工运行。当液晶显示提问 LP. OR P.? 时,请回答 LP (表示计算结果打印出来) 或 P (只显示不打印)。如回答 P 还应规定显示时间,详见 PRO 1-1-02 的操作指南。

(四) 如果 A 阵退化,程序将显示无逆标志 MATRIX SINGULAR!! ,并结束运行。

本例回答 LP, 可从窄条上获得计算结果 A^{-1} , 它以 2 号字排列打印, 详见打印样品。注意 A^{-1} 的行标志 "0:, 1:, 2:, ……" 以及列标志 "0, 1, 2, ……" , 可用不同颜色打印, 以便

$$A^{-1} = \begin{pmatrix} -0.77 & -0.11 & 0.33 \\ 1.22 & 0.88 & -0.66 \\ -0.11 & -0.44 & 0.33 \end{pmatrix}$$

与逆矩阵元素相区别。

打印样品：

	0	1	2
0:	-0.77	-0.11	0.33
1:	1.22	0.88	-0.66
2:	-0.11	-0.44	0.33

三、程序清单

```

5 REM PRO 1-1-03
6 REM PC--1500 " A^-1 "
10 CLEAR
20 READ N
30 DIM A(N-1,N-1),V(N-1)
40 FOR I=0 TO N-1:FOR J=0 TO N-1
50 READ A(I,J)
60 NEXT J:NEXT I
70 FOR L=0 TO N-1:FOR I=L TO N-1
90 IF A(I,L)<>0 GOTO 120
100 NEXT I
110 BEEP 20:PRINT "MATRIX SINGULAR !!":GOTO 600
120 GOSUB "L1"
130 A(L,L)=1/A(L,L)
140 GOSUB "L2"
150 FOR I=0 TO N-1
160 IF I=L GOTO 180
170 B=A(I,L):A(I,L)=0
175 GOSUB "L3"
180 NEXT I
190 NEXT L
200 FOR L=N-1 TO 0 STEP -1
210 IF V(L)=L GOTO 250
220 FOR I=0 TO N-1
230 B=A(I,L):A(I,L)=A(I,V(L)):A(I,V(L))=B
240 NEXT I
250 NEXT L:GOTO "DD"
260 "L1":IF I=L GOTO 290
265 FOR J=0 TO N-1
270 B=A(L,J):A(L,J)=A(I,J):A(I,J)=B
280 NEXT J
290 V(L)=I
300 RETURN
310 "L2":FOR J=0 TO N-1

```

```

320 IF J=L GOTO 340
330 A(L, J)=A(L, L)*A(L, J)
340 NEXT J
350 RETURN
360 "L3":FOR J=0 TO N-1
370 A(I, J)=A(I, J)-B*A(L, J)
380 NEXT J
385 RETURN
390 "DD":INPUT "LP. OR P. ?":S$
395 IF S$="P" GOTO 550
400 INPUT "CSIZE?":CS
405 GRAPH:ROTATE 1:CSIZE CS
410 US$="#####":US$=US$+"#":WS$="###.##"
415 CN=24/CS
420 P=INT (N/CN):PL=N-P*CN
430 P=P+SGN (PL)
440 FOR PA=1 TO P
445 Y=6*CS*(LEN WS$+1)
450 LINE (0,0)-(210,0),8,3
455 K=CN:IF (PL>0)*(PA=P)LET K=PL
460 FOR J=1 TO N
465 FOR I=0 TO K
470 X=(CN-I)/CN*200
475 T=CN*(PA-1)+I
480 IF (J=1)*(I>0)GLCURSOR (X,0):COLOR 2:LPRINT USIN
G US$:T-1,"I"
485 IF I=0 LET E=J-1:COLOR 3:USING US$:GOTO 495
490 E=A(T-1, J-1):COLOR 2:USING WS$
495 GLCURSOR (X,-Y):LPRINT E
500 NEXT I
505 IF J=NLET Y=2*Y
510 GLCURSOR (0,-Y):SORGN
520 NEXT J
530 NEXT PA
535 LINE (0,-Y/4)-(210,-Y/4),8,3
540 TEXT:LF 3:GOTO 600
550 INPUT "TIME?":T
560 WAIT 64*T
570 FOR I=0 TO N-1:FOR J=0 TO N-1
580 PRINT "A(" +STR$(I)+", "+STR$(J)+")=":A(I, J)
590 NEXT J:NEXT I
600 END
1000 DATA 3
1010 DATA 0,1,2,3,2,1,4,3,5

```

程序名称：迭代法解线性方程组

文 件 名：PRO 1-1-04 占用字节：610

编制者：宛公展

一、功能与方法概述

设待解线性方程组为

$$AX = B \quad (1)$$

由(1)式可导出如下迭代形式，即

$$X = C^{-1} (B - (A - C) X) \quad (2)$$

式中C为A矩阵对角线元素组成的对角矩阵，即 $C = \text{diag}(a_{11}, a_{22}, \dots, a_{nn})$ 。(2)式也可写成

$$X_i = \left(b_i - \sum_{j=1}^n a_{ij} X_j \right) / a_{ii} \quad (i, j=1, 2, \dots, n; j \neq i) \quad (3)$$

在(3)式中, 先给 X_j 一组初值, 可解出一组 X_i , 再将 X_i 赋值给 X_j , 代入(3)式又可得到一组新的 X_i , 如此迭代下去, 直到 $|X_i - X_j| \leq EPS$ 为止。这里的 EPS 为规定的误差控制标准。 X_i 即为方程组(1)的近似解。

二、操作指南与举例

设方程组(1)系数阵的增广阵为 $\begin{pmatrix} 4 & 1 & -2 & 9 \\ 1 & 3 & -1 & 4 \\ 2 & -5 & 4 & 9 \end{pmatrix}$ 求 $X_i = ? (i=1, 2, 3)$, 操作步骤如下:

步骤如下:

(一) 在程序的 500 句, 置入待解方程组的未知数个数与迭代误差控制。本例 $N=3$, $EPS=10^{-8}$ 。

(二) 在 510 句, 将增广矩阵按行置入读语句区。

(三) 用 RUN 指令开工运行, 当液晶显示 $X(1)=?$ $X(2)=?$, ..., $X(n)=?$ 时, 请给 X_i 赋一组初值, 通常可取 $0, 0, \dots, 0$ 。

(四) 本例运行结果, 见打印样品。其中 $X_1, X_2, \dots, X_n$ 为初值; NUM 为迭代次数; E(MAX) 为迭代最大误差值; $X(1), X(2), X(3)$ 为方程的根。本样品中包含赋不同初值的运算结果。其中一组的初值为 $0, 0, 0$; 另一组初值为 $3, 1, 2$ 。

打印样品:

```
X1, X2, ... Xn= 0 0 0
NUM=48
E(MAX) 0.000000008

X(1)= 3.000000003
X(2)= 1.000000003
X(3)= 1.999999998
```

```
X1, X2, ... Xn= 3 1 2
NUM=0
E(MAX) 0
```

```
X(1)= 3
X(2)= 1
X(3)= 2
```

三、程序清单

```
5 REM PRO 1-1-04
6 REM " A*X=B (ITERATION)"
10 CLEAR
20 READ N, EPS
```

```

25 DIM A(N,N+1),X(N),Y(N)
30 FOR I=1 TO N:FOR J=1 TO N+1
40 READ A(I,J)
50 NEXT J:NEXT I
60 NU=0:WAIT 0
70 FOR I=1 TO N
80 CLS :PRINT "X"+STR$(I)+"=":";INPUT X(I)
85 NEXT I:CSIZE 1
90 LPRINT "X1,X2,...Xn=":
95 FOR I=1 TO N
100 LPRINT X(I):
105 NEXT I:LPRINT :LPRINT :CLS
110 FOR I=1 TO N
115 K=A(I,N+1)
120 FOR J=1 TO N
125 IF J=IGOTO 140
130 K=K-A(I,J)*X(J)
140 NEXT J
150 Y(I)=K/A(I,I)
160 NEXT I
170 MA=0:FOR I=1 TO N
180 D=ABS (Y(I)-X(I)):X(I)=Y(I)
190 IF D>MALET MA=D
200 NEXT I:CSIZE 2
210 IF MA<EPSGOTO 230
220 NU=NU+1:PRINT NU:GOTO 110
230 LPRINT "NUM=":STR$(NU)
240 LPRINT "E(MAX):":MA:LPRINT
250 FOR I=1 TO N
260 LPRINT "X("+STR$(I)+")=":X(I)
270 NEXT I:LF 3
280 END
500 DATA 3,1E-8
510 DATA 4,1,-2,9,1,3,-1,4,2,-5,4,9

```

程序名称：高斯-约当消去法解线性方程组

文件 名：P R O 1-1-05 占用字节：1062

编制者：宛公展

一、功能与方法概述

本程序先对待解方程系数矩阵的主元进行扫描，如果当步的主元为0，程序将沿列找到一个非0的元素，然后作行交换，使主元变为非0，再用高斯-约当(Gauss-Jordan)消去法实施如下变换：

$$a_{ij} = \begin{cases} a_{ij}/a_{ii} & (\text{当 } i = k \text{ 时}) \\ a_{ij} - a_{ik} a_{kj} & (i \neq k) \end{cases}$$

这里，k 为变换的步数(k=1,2,...,n)。A的零步阵，即为方程组系数矩阵的增广矩阵。如果系数矩阵退化，程序将显示无解标志。

二、操作指南与举例

设待解方程的系数增广矩阵为

$$A = \begin{pmatrix} 0 & 0 & -2 & -6.90 \\ 2 & -3 & 0 & -4.56 \\ 3 & 1 & 0 & 6.03 \end{pmatrix}$$

求出方程的根。

(一) 在程序的1000句置入方程组未知数个数N。本例N = 3。

(二) 从1010句开始按行将增广阵A置入读语句区。

(三) 用RUN指令开工运行。当液晶显示LP. ORP.? 时, 请回答LP (打印计算结果) 或P (只显示计算结果)。若回答LP, 计算机再提问CSIZE(1 OR 2)? 时, 还要键入打印字号, 可回答1或2。

(四) 本例求解结果, 详见打印样品。样品为增广阵的最后一步变换结果, 它的前三列已变为单位矩阵, 最后一列依次为方程的根 X_1 , X_2 , X_3 。

打印样品:

	1	2	3	4	
1.00	0.00	0.00	1.23	=X1	
0.00	1.00	0.00	2.34	=X2	
0.00	0.00	1.00	3.45	=X3	

三、程序清单

```

5 REM PRO 1-1-05
6 REM "A*X=B" (GAUSS-JORDAN)
10 CLEAR
20 READ N
30 DIM A(N,N+1)
40 FOR I=1 TO N:FOR J=1 TO N+1
50 READ A(I,J)
60 NEXT J:NEXT I
70 FOR L=1 TO N:FOR I=L TO N
90 IF A(I,L)<>0 GOTO 120
100 NEXT I
110 BEEP 20:PRINT "NOT ROOT !!":GOTO 550
120 IF I=L GOTO 170
130 FOR J=1 TO N+1
140 B=A(L,J):A(L,J)=A(I,J):A(I,J)=B
150 NEXT J:BEEP 1,200,30
170 C=A(L,L)
180 FOR J=1 TO N+1
190 A(L,J)=A(L,J)/C
200 NEXT J
210 FOR K=1 TO N
220 IF K=L GOTO 270
230 U=A(K,L)
240 FOR J=L TO N+1

```

```

250 A(K, J)=A(K, J)-U*A(L, J)
260 NEXT J
270 NEXT K
280 NEXT L
290 INPUT "LP. OR P. ?":S$
300 IF S$="LP" GOTO 400
310 FOR I=1 TO N
320 CLS :PRINT "X "+STR$(I)+"=" :A(I, N+1)
330 NEXT I :GOTO 550
400 INPUT "CSIZE(1 OR 2)?":CS
405 GRAPH :ROTATE 1 :CSIZE CS
410 U$="#####":W$="####.##"
415 CN=24/CS
420 P=INT (N/CN):PL=N-P*CN
430 P=P+SGN (PL)
440 FOR PA=1 TO P
445 Y=6+CS*(LEN W$+1)
450 LINE (0, -Y/2)-(210, -Y/2), 8, 3
455 K=CN:IF (PL>0)*(PA=P) LET K=PL
460 FOR J=1 TO N+1
465 FOR I=0 TO K
470 X=(CN-I)/CN*200
475 T=CN*(PA-1)+I
485 IF I=0 LET E=J:COLOR 3:USING U$:GOTO 495
490 E=A(T, J):COLOR 2:USING W$
495 GLCURSOR (X, -Y):LPRINT E
496 IF (PA=P)*(J=N+1)*(I>0)GLCURSOR (X, -2*Y):LPRINT
"=X":STR$(T)
500 NEXT I
505 IF J=N+1 LET Y=2*Y
510 GLCURSOR (0, -Y):SORGN
520 NEXT J
530 NEXT PA
535 LINE (0, -Y/2)-(210, -Y/2), 8, 3
540 TEXT :LF 5
550 END
1000 DATA 3
1010 DATA 0, 0, -2, -6.9, 2, -3, 0, -4.56, 3, 1, 0, 6.03

```

程序名称: 求实对称矩阵的特征值与特征向量

文件 名: PRO 1-1-06 占用字节: 1156

编制者: 宛公展

一、功能与方法概述

设 n 阶实对称方阵 A 的特征问题可用下式表示, 即

$$AX = X\Lambda \quad (1)$$

在矩阵 X 中, 按列存放 A 的每个特征向量, Λ 为存放特征值的对角矩阵。为求解 A 的特征问题, 可对矩阵 A 用 Jacobi 方法实施正交旋转变化, 即

$$A_k = R_k \cdot A_{k-1} \cdot R_k' \quad (2)$$

这里, $k=1, 2, \dots, r$ 表示变化的步数。 R_k 为第 k 步左乘的正交旋转阵。当进行了有限次 r 的旋转变化后, 可使 A 阵转化为对角型。此时, 该阵对角线上的元素, 即为相应的特征值。为加快收敛速度, 通常在 A 阵的上三角中, 选取一个模数最大的元素, 记为 a_{pq} ($p < q$),

则 R_k 阵的 p 行 p 列, q 行 q 列可记为 $\begin{pmatrix} \cos\theta & \sin\theta \\ -\sin\theta & \cos\theta \end{pmatrix}$,其余元素,同于单位矩阵。另外,

这里的 θ 应满足 $\tan 2\theta = 2a_{pq}/(a_{pp}-a_{qq})$ 。(2)式对应如下变换:

$$\begin{cases} a_{ip}^{(k)} = a_{ip}^{(k-1)} \cos\theta + a_{iq}^{(k-1)} \sin\theta \\ a_{iq}^{(k)} = a_{iq}^{(k-1)} \cos\theta - a_{ip}^{(k-1)} \sin\theta \\ (i=1,2,\dots,n; i \neq p, i \neq q) \\ a_{pp}^{(k)} = a_{pp}^{(k-1)} \cos^2\theta + a_{qq}^{(k-1)} \sin^2\theta + a_{pq}^{(k-1)} \sin 2\theta \\ a_{qq}^{(k)} = a_{pp}^{(k-1)} \sin^2\theta + a_{qq}^{(k-1)} \cos^2\theta - a_{pq}^{(k-1)} \sin 2\theta \\ a_{pq}^{(k)} = 0 \end{cases} \quad (3)$$

至于特征向量 X ,可在旋转中逐步形成,即

$$X_k = X_{k-1} \cdot R_k$$

这里, $k=1, 2, \dots, r$,仍表示变化的步数; X 的零步阵,应取为单位矩阵 I 。

二、操作指南与举例

例如,求矩阵 $A = \begin{pmatrix} 2 & -1 & 0 \\ -1 & 2 & -1 \\ 0 & -1 & 2 \end{pmatrix}$ 的特征值与特征向量。操作步骤如下:

(一) 在程序的500句,置入矩阵 A 的阶数 N 与迭代误差控制 EPS 。本例 $N=3$, $EPS=10^{-20}$

(二) 从510句开始按行将矩阵 A 的元素置入读语句区。

(三) 用 RUN 指令开工运行后,液晶板上自动显示迭代次数标志,当方阵 A 达到 EPS 控制的对角型时,将结果打印出来。请见打印样品,这里 NUM 为迭代次数, MA 为 A 阵上三角区域最大元素(不包含对角线元素)数值,其余分段打印的各组数据中,第一行为特征值(LAMDA 1,即为特征值 $\lambda_1, \dots$),余下的为特征向量的数值。

打印样品:

```
NUM:11  MAX:0
LAMDA 1:  3.41E 00
```

```
-5.000000189E-01
 7.071067622E-01
-0.500000000E-01
```

```
LAMDA 2:  2.00E 00
```

```
 7.071076586E-01
 1.251735266E-06
-7.071059039E-01
```

```
LAMDA 3:  5.85E-01
```

```
-4.999987402E-01
-7.071068004E-01
-5.000012328E-01
```

三、程序清单

```

5 REM PRO 1-1-06
6 REM A*X=X*Diag(^) "JACOB"
10 CLEAR :WAIT 0
20 READ N, EPS
30 DIM A(N, N), P(N, N)
40 FOR I=1 TO N
45 P(I, I)=1
50 FOR J=1 TO N
55 READ A(I, J)
60 NEXT J:NEXT I
65 L=1
70 "S":MA=0:FOR I=1 TO N-1:FOR J=I+1 TO N
80 IF ABS A(I, J)>ABS MALET P=I:Q=J:MA=A(P, Q)
90 NEXT J:NEXT I
95 IF ABS MA<EPSGOTO "DD"
100 E=A(P, P)-A(Q, Q):F=A(P, P)+A(Q, Q)
105 D=SQR (E*E+4*MA*MA)
110 IF E=0 LET C=SQR 2/2:S=SGN (MA)*C:GOTO 135
120 C=SQR ((1+ABS (E)/D)/2)
130 S=SGN (MA*E)*SQR ((1-ABS (E)/D)/2)
135 IF ABS A(P, P)<=ABS A(Q, Q)LET E=S:S=-C:C=E
140 FOR K=1 TO N
150 IF (K=P)OR (K=Q)GOTO 290
160 IF K<PGOTO 260
170 IF K<QGOTO 220
180 PK=A(P, K):QK=A(Q, K)
190 A(P, K)=C*PK+S*QK
200 A(Q, K)=-S*PK+C*QK
210 GOTO 290
220 PK=A(P, K):KQ=A(K, Q)
230 A(P, K)=C*PK+S*KQ
240 A(K, Q)=-S*PK+C*KQ
250 GOTO 290
260 KP=A(K, P):KQ=A(K, Q)
270 A(K, P)=C*KP+S*KQ
280 A(K, Q)=-S*KP+C*KQ
290 NEXT K
300 PP=A(P, P):QQ=A(Q, Q)
310 A(P, P)=PP*C+C*QQ*S*S+2*MA*S*C
320 A(Q, Q)=PP*S+S*QQ*C+C-2*MA*S*C
330 A(P, Q)=0
340 FOR I=1 TO N
350 IP=P(I, P):IQ=P(I, Q)
360 P(I, P)=IP*C+IQ*S
370 P(I, Q)=-IP*S+IQ*C
380 NEXT I
390 PRINT L:IF L/5=INT (L/5)BEEP 1
391 L=L+1:GOTO "S"
400 "DD":C$IZE 1:COLOR 2
405 LPRINT "NUM:":STR$ (L):" MA=":MA
410 C$IZE 2:LPRINT
415 FOR J=1 TO N
425 LPRINT USING "#.##^":"LAMDA "+STR$ (J)+": ":A(J,
J)
430 LPRINT :USING
435 FOR I=1 TO N
440 LPRINT P(I, J)
450 NEXT I:LPRINT
460 NEXT J
470 LF 3
480 END
500 DATA 3, 1E-20
510 DATA 2, -1, 0, -1, 2, -1, 0, -1, 2

```